'Set clock for 8 MHz (Atmega328)

Config PortB = output                     'PortB is used to drive the Fail and Charging LEDs and the Charger Relay

Config PortD = output                     'PortD is used to drive the Cell Relays

Config PortC.5 = output                   'PortC.5 is used to drive the Charge Complete LED

Config PortC.4 = input                    'PortC.4 is used to monitor the Cell Voltages

Dim Voltagelowbyte as Byte                'ADC low value byte

Dim Voltagehighbyte as Byte               'ADC high value byte

Dim Voltage as Word                       'ADC combined value word

Dim N as Integer                         'Used for the for - next - loop

Dim V as Single                           'V is the calculated cell voltage

Dim V1 as Single                          'V1 t0 V4 are memory values of the four cell voltages

Dim V2 as Single

Dim V3 as Single

Dim V4 as Single

Dim CellFullThreshold as Single           'Cell Voltage at Full

Dim CellTooLow as Single                  'Minimum allowable cell voltage

Dim CellTooHigh as Single                 'Maximum allowable cell voltage

portB.2 = 0       'Switch Charger Relay off               'Zeroize all values

Portb.1=0         'Switch Charge Indicator Off

Portb.0=0         'Switch Fail Indicator Off

PortC.5=0         'Switch Charge Complete Indicator Off

Portd = 0         'Switch Cell Relays Off

V = 0

V1 = 0

V2 = 0

V3 = 0

N = 0

Voltage = 0

Voltagehighbyte = 0

Voltagelowbyte = 0

CellFullThreshold = 3.5/5.0                   'Calculate threshold voltages

CellFullThreshold = CellFullThreshold * 1024   'These voltages are dependent on the 5V regulator actual voltage

                                               'Cell voltage at full is set for 3.5V

CellTooLow = 2.5/5.0                          'Cell voltage minimum is set for 2.5V

CellTooLow = CellTooLow * 1024

CellTooHigh = 3.6/5.0                         'Cell voltage maximum is set for 3.6V

CellTooHigh = CellTooHigh * 1024

Out &H7C , &B00000100                                    'Setup the Adc - Port ADC4

Out &H7A , &B11010111

Waitms 1

   Voltagelowbyte = Inp(&H78)                               'Read Cell Voltage

   Voltagehighbyte = Inp(&H79)

   Voltage = Makeint(Voltagelowbyte , Voltagehighbyte) 'Combine ADC bytes into a word

If Voltage > &B1100000 then goto Fault        'Check that cell switches are not stuck in on position

wait 45    'Ajust with 5 for each processor, 1 to 10  - processor 1 is wait 1 and processor 10 is wait 45

Start:

                                              'Main Routine measures four cell voltages

For N = 1 to 4                                'and decide action, proceed or stop.

PORTD.N=1

Wait 4

Out &H7C , &B00000100                                    'Setup the Adc - Port ADC4

Out &H7A , &B11010111

   Waitms 1

   Voltagelowbyte = Inp(&H78)                               'Read Cell Voltage

   Voltagehighbyte = Inp(&H79)

   Voltage = Makeint(Voltagelowbyte , Voltagehighbyte) 'Combine ADC bytes into a word

If N = 1 Then V1 = Voltage * 1              'Calculate and store cell voltages

If N = 2 Then V2 = Voltage * 2

If N = 3 Then V3 = Voltage * 3

If N = 4 Then V4 = Voltage * 4

If N = 1 Then V = V1                        'Calculate cell voltage of each cell to compare with threshold

If N = 2 Then V = V2 - V1                   'voltages

If N = 3 Then V = V3 - V2

If N = 4 Then V = V4 - V3

If V < CellTooLow then goto Terminate          'Lower than 2V

If V >= CellFullThreshold then goto Terminate  'Higher than 3.5V

portb.2 = 1      'Switch Charger Relay On

portb.1 = 1      'Indicate Charge Mode On

PORTD.N=0

Wait 1

Out &H7C , &B00000100                                    'Setup the Adc - Port ADC4

Out &H7A , &B11010111

Waitms 1

   Voltagelowbyte = Inp(&H78)                               'Read Cell Voltage

   Voltagehighbyte = Inp(&H79)

   Voltage = Makeint(Voltagelowbyte , Voltagehighbyte) 'Combine ADC bytes into a word

If Voltage > &B1100000 then goto Fault        'Check that cell switches are not stuck in on position

Next N

Goto Start

Fault:       'Stop due to cell relays stuck and indicate this condition

portb.1 = 0      'Indicate Charge Mode Off

portb.2 = 0      'Switch Charger Relay Off

Portd = 0        'Switch Cell Relays Off

portb.0 = 1

Wait 1

portb.0 = 0

Wait 1

Goto Fault

Terminate:       'Decide to stop as failure (cell voltages too high or too low) or indicate charge complete

portb.1 = 0      'Indicate Charge Mode Off

portb.2 = 0      'Switch Charger Relay Off

Portd = 0        'Switch Cell Relays Off

If V <= CellTooHigh then portc.5 = 1       'Indicate Charge Complete (smaller than 3.6V)

If V > CellTooHigh then portb.0 = 1        'Indicate Fail when Cell Voltage bigger than 3.6V

If V > CellTooHigh then portC.5 = 0        'Switch off Charge Complete Indicator

If V < CellTooLow then portb.0 = 1         'Indicate Fail when Cell Voltage lower than 2.0V

If V < CellTooLow then portC.5 = 0         'Switch off Charge Complete Indicator

End

